Maxwell equations (first set of equations)

Let us introduce the following Ore algebra A of partial differential operators in four variables t, x_1, x_2, x_3.

```
A = OreAlgebra[Der[t], Der[x1], Der[x2], Der[x3]]
```

$\mathbb{K}(t, x_1, x_2, x_3)[D_t; 1, D_t][D_{x_1}; 1, D_{x_1}][D_{x_2}; 1, D_{x_2}][D_{x_3}; 1, D_{x_3}]$

and the matrix R of partial differential operators, which defines the following linear system

```
R = {{Der[t], 0, 0, 0, -Der[x3], Der[x2]},
    {0, Der[t], 0, Der[x3], 0, -Der[x1]},
    {0, 0, Der[t], -Der[x2], Der[x1], 0},
    {Der[x1], Der[x2], Der[x3], 0, 0, 0}};
MatrixForm[R = ToOrePolynomial[R, A]]
```

$$\begin{pmatrix} D_t & 0 & 0 & 0 & -D_{x_3} & D_{x_2} \\ 0 & D_t & 0 & D_{x_3} & 0 & -D_{x_1} \\ 0 & 0 & D_t & -D_{x_2} & D_{x_1} & 0 \\ D_{x_1} & D_{x_2} & D_{x_3} & 0 & 0 & 0 \end{pmatrix}$$

and the left A-module, finitely presented by R, defined by M = $A^{1\times 6} / A^{1\times 4} R$.

Let us write the equations of the system explicitly:

```
u = {t, x1, x2, x3};
EMF = {B1 @@ u, B2 @@ u, B3 @@ u, E1 @@ u, E2 @@ u, E3 @@ u}
```

$\{B1[t, x_1, x_2, x_3], B2[t, x_1, x_2, x_3], B3[t, x_1, x_2, x_3], \mathcal{E}1[t, x_1, x_2, x_3], \mathcal{E}2[t, x_1, x_2, x_3], \mathcal{E}3[t, x_1, x_2, x_3]\}$

```
sys = Thread[ApplyMatrix[R, EMF] == 0]
```

$\{-\mathcal{E}2^{(0,0,0,1)}[t, x_1, x_2, x_3] + \mathcal{E}3^{(0,0,1,0)}[t, x_1, x_2, x_3] + B1^{(1,0,0,0)}[t, x_1, x_2, x_3] == 0,$
$\mathcal{E}1^{(0,0,0,1)}[t, x_1, x_2, x_3] - \mathcal{E}3^{(0,1,0,0)}[t, x_1, x_2, x_3] + B2^{(1,0,0,0)}[t, x_1, x_2, x_3] == 0,$
$-\mathcal{E}1^{(0,0,1,0)}[t, x_1, x_2, x_3] + \mathcal{E}2^{(0,1,0,0)}[t, x_1, x_2, x_3] + B3^{(1,0,0,0)}[t, x_1, x_2, x_3] == 0,$
$B3^{(0,0,0,1)}[t, x_1, x_2, x_3] + B2^{(0,0,1,0)}[t, x_1, x_2, x_3] + B1^{(0,1,0,0)}[t, x_1, x_2, x_3] == 0\}$

Let us compute the adjoint of R:

```
MatrixForm[Radj = Involution[R, A]]
```

$$\begin{pmatrix} -D_t & 0 & 0 & -D_{x_1} \\ 0 & -D_t & 0 & -D_{x_2} \\ 0 & 0 & -D_t & -D_{x_3} \\ 0 & -D_{x_3} & D_{x_2} & 0 \\ D_{x_3} & 0 & -D_{x_1} & 0 \\ -D_{x_2} & D_{x_1} & 0 & 0 \end{pmatrix}$$

Let us check whether or not M is torsion-free:

```
MatrixForm /@ ({Ann, Rp, Q} = Exti[Radj, A, 1])
```

$$\left\{\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}, \begin{pmatrix} 0 & 0 & -D_t & D_{x_2} & -D_{x_1} & 0 \\ 0 & -D_t & 0 & -D_{x_3} & 0 & D_{x_1} \\ -D_{x_1} & -D_{x_2} & -D_{x_3} & 0 & 0 & 0 \\ -D_t & 0 & 0 & 0 & D_{x_3} & -D_{x_2} \end{pmatrix}, \begin{pmatrix} 0 & 0 & -D_{x_3} & D_{x_2} \\ 0 & D_{x_3} & 0 & -D_{x_1} \\ 0 & -D_{x_2} & D_{x_1} & 0 \\ D_{x_1} & -D_t & 0 & 0 \\ D_{x_2} & 0 & -D_t & 0 \\ D_{x_3} & 0 & 0 & -D_t \end{pmatrix}\right\}$$

Since the first matrix is identity, M is torsion-free and thus the first set of Maxwell equations is parametrized by Q, i.e:

```
Thread[EMF == ApplyMatrix[Q, {ξ1 @@ u, ξ2 @@ u, ξ3 @@ u, ξ4 @@ u}]]
```

$$\begin{aligned}\{&B1[t, x_1, x_2, x_3] == -\xi_3^{(0,0,0,1)}[t, x_1, x_2, x_3] + \xi_4^{(0,0,1,0)}[t, x_1, x_2, x_3], \\ &B2[t, x_1, x_2, x_3] == \xi_2^{(0,0,0,1)}[t, x_1, x_2, x_3] - \xi_4^{(0,1,0,0)}[t, x_1, x_2, x_3], \\ &B3[t, x_1, x_2, x_3] == -\xi_2^{(0,0,1,0)}[t, x_1, x_2, x_3] + \xi_3^{(0,1,0,0)}[t, x_1, x_2, x_3], \\ &\mathcal{E}1[t, x_1, x_2, x_3] == \xi_1^{(0,1,0,0)}[t, x_1, x_2, x_3] - \xi_2^{(1,0,0,0)}[t, x_1, x_2, x_3], \\ &\mathcal{E}2[t, x_1, x_2, x_3] == \xi_1^{(0,0,1,0)}[t, x_1, x_2, x_3] - \xi_3^{(1,0,0,0)}[t, x_1, x_2, x_3], \\ &\mathcal{E}3[t, x_1, x_2, x_3] == \xi_1^{(0,0,0,1)}[t, x_1, x_2, x_3] - \xi_4^{(1,0,0,0)}[t, x_1, x_2, x_3]\}\end{aligned}$$

We note the above parametrization involves four arbitrary functions ξ_1, ... ξ_4. Let us now compute the rank of the system

```
OreRank[R, A]
```

3

From that we know that there exist parametrizations involving only three arbitrary functions. Let us compute some of them:

```
MatrixForm /@ (paras = MinimalParametrizations[R, A])
```

$$\left\{\begin{pmatrix} 0 & 0 & -D_{x_3} \\ 0 & D_{x_3} & 0 \\ 0 & -D_{x_2} & D_{x_1} \\ D_{x_1} & -D_t & 0 \\ D_{x_2} & 0 & -D_t \\ D_{x_3} & 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & 0 & D_{x_2} \\ 0 & D_{x_3} & -D_{x_1} \\ 0 & -D_{x_2} & 0 \\ D_{x_1} & -D_t & 0 \\ D_{x_2} & 0 & 0 \\ D_{x_3} & 0 & -D_t \end{pmatrix}, \begin{pmatrix} 0 & -D_{x_3} & D_{x_2} \\ 0 & 0 & -D_{x_1} \\ 0 & D_{x_1} & 0 \\ D_{x_1} & 0 & 0 \\ D_{x_2} & -D_t & 0 \\ D_{x_3} & 0 & -D_t \end{pmatrix}, \begin{pmatrix} 0 & -D_{x_3} & D_{x_2} \\ D_{x_3} & 0 & -D_{x_1} \\ -D_{x_2} & D_{x_1} & 0 \\ -D_t & 0 & 0 \\ 0 & -D_t & 0 \\ 0 & 0 & -D_t \end{pmatrix}\right\}$$

For example, let us consider the first one.

```
sol = Thread[EMF == ApplyMatrix[paras[[1]], {ξ1 @@ u, ξ2 @@ u, ξ3 @@ u}]]
```

$$\begin{aligned}\{&B1[t, x_1, x_2, x_3] == -\xi_3^{(0,0,0,1)}[t, x_1, x_2, x_3], \\ &B2[t, x_1, x_2, x_3] == \xi_2^{(0,0,0,1)}[t, x_1, x_2, x_3], \\ &B3[t, x_1, x_2, x_3] == -\xi_2^{(0,0,1,0)}[t, x_1, x_2, x_3] + \xi_3^{(0,1,0,0)}[t, x_1, x_2, x_3], \\ &\mathcal{E}1[t, x_1, x_2, x_3] == \xi_1^{(0,1,0,0)}[t, x_1, x_2, x_3] - \xi_2^{(1,0,0,0)}[t, x_1, x_2, x_3], \\ &\mathcal{E}2[t, x_1, x_2, x_3] == \xi_1^{(0,0,1,0)}[t, x_1, x_2, x_3] - \xi_3^{(1,0,0,0)}[t, x_1, x_2, x_3], \\ &\mathcal{E}3[t, x_1, x_2, x_3] == \xi_1^{(0,0,0,1)}[t, x_1, x_2, x_3]\}\end{aligned}$$

```
ApplyMatrix[R, ApplyMatrix[paras[[1]], {ξ1 @@ u, ξ2 @@ u, ξ3 @@ u}]]
```

{0, 0, 0, 0}